OEIS A175497: PRODUCTS OF TRIANGULAR NUMBERS WHICH ARE PERFECT SQUARES.

RICHARD J. MATHAR

ABSTRACT. Which products of two distinct triangular numbers are perfect squares? Sequence [3, A175497] is a list of the bases of these perfect squares. We provide an algorithm which relates a known index of one of the triangular factors to a Pell equation to find the indices of the other triangular factors.

1. NOTATION

The triangular numbers are [3, A000217]

$$T(n) \equiv \frac{n(n+1)}{2} \tag{1}$$

and products of triangular numbers may be perfect squares

$$T(n)T(m) \stackrel{?}{=} s^2 \tag{2}$$

with bases s collected in [3, A175497]. The trivial solutions are $n = m$ which define the square triangular numbers $T(n)^2$ [3, A000537]. So the principle question asks for solutions in the index range $m < n$.

Remark 1. *There are cases where s^2 is a square triangular number but also a product of* distinct *triangular number, e.g.* $T(3)^2 = T(1)T(8) = 6^2$.

Multiplying (2) by 4 we target the equivalent question for products of oblong numbers [3, A002378] being even squares,

$$n(n+1)m(m+1) \stackrel{?}{=} (2s)^2. \tag{3}$$

2. ALGORITHM

Given an upper search limit for s, the task is to scan solutions of (3) for each n in the range $2n \le 1 + \sqrt{1+8s^2}$. So we consider n given and search for the set of matching m. Our key idea is the observation that the oblong $n(n+1)$ has a unique prime factorization according to the fundamental law of algebra; to construct a perfect square $(2s)^2$, $m(m+1)$ must complement the prime factors with odd exponents such that the sum of both exponents of each prime factor in $n(n+1)m(m+1)$ becomes even; apart from that requirement $m(m+1)$ may be multiplied by any other perfect square. So the requirement is that the squarefree parts of $n(n+1)$ and $m(m+1)$ are equal [3, A007913]:

$$\text{core}(n(n+1)) = \text{core}(m(m+1)). \tag{4}$$

Date: March 16, 2023.
2020 *Mathematics Subject Classification.* Primary 11A05, 11A07.

Remark 2. *The squarefree part is a multiplicative arithmetic function and* n *and* $n+1$ *are coprime, so* $\mathrm{core}(n(n+1)) = \mathrm{core}(n)\,\mathrm{core}(n+1)$ [3, A083481].

Remark 3. *The number of distinct prime factors in the squarefree part of the* n*-th oblong number is*

$$\omega(\mathrm{core}(n^2+n)) = 1,2,1,1,3,3,2,1,2,3,2,2,3,4,2,1,2,2,2,3,4,\ldots, n \ge 1. \tag{5}$$

Given n, the radix

$$r \equiv \mathrm{core}(n^2+n) \tag{6}$$

and the perfect square

$$\square_n \equiv \frac{n(n+1)}{r} \tag{7}$$

are also known [3, A083481]:

$$r = 2,6,3,5,30,42,14,2,10,110,33,39,182,210,15,17\ldots, n \ge 1. \tag{8}$$

The solutions of (3) require that the oblong factor is r times a perfect square:

$$m(m+1) \stackrel{!}{=} ry^2. \tag{9}$$

The usual approach for diophantine equations is to diagonalize the quadratic form on the left hand side, $m(m+1) = (m+1/2)^2 - 1/4$, so the task is to solve the "classical" Pell Equation

$$(2m+1)^2 - 4ry^2 = 1. \tag{10}$$

Remark 4. *Distinct indices* n *may lead to the same* r*, so there may be duplicates in the set of* s *that are generated by looping over distinct* n*.*

The standard approach is to start with the continued fraction expansion [1] of

$$D \equiv 4r \tag{11}$$

to find the fundamental solution of

$$x^2 - Dy^2 = 1 \tag{12}$$

[5, 4, 10, 15, 7, 8, 6] and to compute the powers of the associated quadratic surd to find the general solution [5, Cor. 1.10][12].

Because the complete factorization of D is known, it suffices to solve

$$x^2 - r(2y)^2 = 1 \tag{13}$$

because the solutions x in (12) and (13) are the same and the solutions y differ only by a factor 2 [5, p. 16].

Remark 5. *The period lengths of the squarefree integers are listed in* [3, A035015]. *The period lengths of squarefree parts of the oblong numbers are*

$$1,2,2,1,2,2,4,1,1,2,4,2,2,2,2,1,4,2,4,2,2,2,4,2,1,4,6,2,\ldots, n \ge 1. \tag{14}$$

If a series of solutions/convergents of (13) is obtained, only the solutions with odd x are kept to match the requirement in (10) that $x = 2m+1$.

Remark 6. *As our* r *are nonsquare, the continued fractions are periodic and palindromic* [9][14].

The trivial solution mentioned in Section 1 means a solution ($x = 2n+1, y^2 = \square_n$) of (10) is already known. This solution $2n+1$ is typically also the fundamental solution, but for $n = 8, 24, 48, 49, 80, 120\ldots$ smaller solutions exist [3, A306415], which means nontrivial solutions of (3) arise. This subset of parameters n arises because the squarefree parts r in the list (8) are not unique functions of n but may show up again later in the list.

Example 1. *$r(1) = 2$ also appears at at $r(8) = r(49) = r(288) = \ldots = 2$. $r(2) = 6$ also appears at $r(24) = r(242) = \cdots = 6$. $r(3) = 3$ also appears at $r(48) = r(675) = \cdots = 3$. $r(4) = 5$ also appears at $r(80) = r(1444) = \cdots = 5$.*

Once a $r(n)$ has been computed, searching backwards in that list (8) for the same r provides a list of smaller indices (i.e. associated m) which have solved equation (10).

Remark 7. *Lenstra writes [4]: "There is no known polynomial time algorithm for deciding whether a given power product represents the fundamental solution to Pell's equation." So apparently building such a list of $r(n)$ values is the only efficient way of predicting where in the (n, m) symmetric square grid of solutions (besides the diagonal of the trivial solutions) these extra m appear.*

The standard theory of continued fractions tells that once a fundamental solution x (resp. m) of the Pell equation is found, the other solutions obey linear recurrences with constant coefficients (C-recurrences) with respect to smaller solutions.

For small n these lists of fundamental solutions m_1 and larger solutions m_i from higher powers in the quadratic Field of $\sqrt{r}$ look as follows:

n	r	m_1	m_2	m_3	m_4	OEIS	recur.
1	2	1	8	49	288	A001108	$m_i = 7(m_{i-1} - m_{i-2}) + m_{i-3}$
2	6	2	24	242	2400	A132596	$m_i = 11(m_{i-1} - m_{i-2}) + m_{i-3}$
3	3	3	48	675	9408	A007654	$m_i = 15(m_{i-1} - m_{i-2}) + m_{i-3}$
4	5	4	80	1444		A132584	$m_i = 29(m_{i-1} - m_{i-2}) + m_{i-3}$
5	30	5	120	2645		A322707	$m_i = 23(m_{i-1} - m_{i-2}) + m_{i-3}$
6	42	6	168	4374		A322708	$m_i = 27(m_{i-1} - m_{i-2}) + m_{i-3}$
7	14	7	224	6727		A322709	$m_i = 31(m_{i-1} - m_{i-2}) + m_{i-3}$
8	2						see $n = 2$
9	10	9	360	13689		A132593	$m_i = 39(m_{i-1} - m_{i-2}) + m_{i-3}$
10	110	10	440	18490			$m_i = 43(m_{i-1} - m_{i-2}) + m_{i-3}$
15	15	15	960	59535			$m_i = 63(m_{i-1} - m_{i-2}) + m_{i-3}$

The mechanism at work here is

- whatever the set of divisors in the starting values of the m-recurrences is, a C-recurrence ensures that the non-fundamental solutions "inherit" that set.
- a recurrence of the form $m_i = \alpha(m_{i-1} - m_{i-2}) + m_{i-3}$ ensures $m_i - 1 = \alpha[(m_{i-1} - 1) - (m_{i-2} - 1)] + m_{i-3} - 1$, which means the "cofactor" in the oblong number obeys the same recurrence and also "inherits" divisors of its earlier members. Altogether this ensures that the divisor r of (9) is maintained in all solutions $m(m + 1)$.

There is a strong heuristics that recurrences for the solutions of the Pell equation in the table above are of the shape

$$m_i = (4n + 3)(m_{i-1} - m_{i-2}) + m_{i-3}. \tag{15}$$

rooted at $m_1 = n$ the fundamental solutions (from the trivial solution, with the exceptions as discussed above) and $m_2 = 4n(n+1)$, $m_3 = n(4n+3)^2$.

Remark 8. *A "telescoping" step allows to rewrite this as inhomogeneous C-recurrences* [3, A322699]

$$m_i = (4n+2)m_{i-1} - m_{i-2} + 2n. \tag{16}$$

The conjectural recurrence is equivalent to the generating functions (GF)

$$\sum_{i\ge 0} m_i x^i = \frac{nx(1+x)}{(1-x)[1-(4n+2)x+x^2]}. \tag{17}$$

Splitting this into partial fractions

$$2\sum_{i\ge 0} m_i x^i = -\frac{1}{(1-x)} + \frac{1-(2n+1)x}{1-2(2n+1)x+x^2} \tag{18}$$

leads to closed form representations with Chebyshev polynomials (denoted by $\hat{T}$ to set them apart from the triangular numbers) [11, 18.12.7][3, A322699]

$$2m_i = \hat{T}_{2n+1}(i) - 1 \tag{19}$$

Looking at the zeros of the denominator of the generating function

$$x^2 - 2(2n+1)x + 1 = 0 \rightsquigarrow x = 2n+1 \pm 2\sqrt{n(n+1)} \tag{20}$$

and an associated Binet-expansion then leads to the type of recurrence expected for the representations of powers of units in the quadratic fields in the theory of the continued fractions [2, 13, Prop. 4.1].

Appendix A. Maple demonstration program

The following is a Maple program (much faster versions exist) where the last line creates a list of all solutions s up to some maximum.

```
#!/usr/bin/env maple

interface(quiet=true):

# @param n
# @return The n-th triangular number
A000217 := proc(n)
        n*(n+1) /2 ;
end proc:


# @param n nonnegative integer
# @return the n-th oblong number
A002378 := proc(n)
        n*(n+1) ;
end proc:

# @param n A nonnegative integer.
# @return squarefree part of n
A007913 := proc(n)
        local f, a, d;
        f := ifactors(n)[2] ;
```

```
        a := 1 ;
        for d in f do
                if type(op(2, d), 'odd') then
                        a := a*op(1, d) ;
                end if;
        end do:
        a;
end proc:

# squarefree part of n-th oblong nubmers
A083481 := proc(n)
        A007913(n)*A007913(n+1) ;
end proc:


# Solve the Pell equation x^2-r*(2*y)^2=1
# where r is the squarefree part of the n-th oblong number
# and x= 2m+1 is odd. Return the fundamental solution [m,y,r].
Pellsolve := proc(r)
        option remember;
        local cf,conv,i,x,y,m ;
        cf := numtheory[cfrac](sqrt(r),'periodic') ;
        for i from 1 do
                conv := numtheory[nthconver](cf,i) ;
                if type(denom(conv),'even') then
                        x := numer(conv) ; # 2*m+1
                        y := denom(conv)/2 ;
                        if  x^2-r*(2*y)^2 = 1 then
                                m := (x-1)/2 ;
                                return [m,y] ;
                        end if;
                end if;
        end do:
        return [0,0] ;
end proc:

# List values of A175497 up to smax
# The values are NOT obtained in sorted order but by fixing an index
# n of the first factor T(n) and considering all T(m), m<n such
# that the product is at most smax.
# [This is a type of CRT scan order in the triangular area where m<n.]
# Also note that the occasional cases where there are two distinct
# factorization for the same square will be printed with multiplicity.
# @param smax an upper limit for the listing of the bases of the squares
# @return The list of solutions s where T(n)*T(m) = s^2, i<> j, s <= smax.
A175497 := proc(smax)
        local n, pellf,pell,m,y,r,itr,twom1,ct,qform,mextra,s ,alls;
        ct := 1 ;
        alls := {0} ;
        for n from 1 do
                # because A000217(m)>=1, no more solutions are
                # found if already this factor passes the maximum.
                if A000217(n) > smax^2 then
```

```
                        break;
                end if;

                r := A083481(n) ;

                # The factor A000217(m) will complement r so
                # n(n+1)m(m+1) is at least n(n+1)*r. Sinc the search
                # is in n(n+1)m(m+1)=(2s)^2, check early that this
                # is within smax: n*(n+1)*r <= 4*s^2
                if n*(n+1)*r > 4*smax^2 then
                        continue ;
                end if;
                # obtain fundamental solution of (2m+1)^2-4*r*y^2=1
                pellf := Pellsolve(r) ;

                m := op(1,pellf) ; y := op(2,pellf) ;
                # if m < n then
                if true then
                        # consider only nontrivial solutions where m < n
                        if m = 0 then
                                print("n=",n,"no pell") ;
                        else
                                qform := 2*m+1+sqrt(4*r)*y ;
                                pell := qform ;
                                for itr from 1 do
                                        # extract 2m+1 without sqrt(r)
                                        twom1 := subs(sqrt(r)=0,pell) ;
                                        mextra := (twom1-1)/2 ;
                                        s := sqrt(n*(n+1)*mextra*(mextra+1))/2 ;
                                        # if mextra < n and s <= smax then
                                        if mextra <> n and s <= smax then
                                                # n*(n+1)*m*(m+1) = n*(n+1)*r*y^2 = (2s)^2 ;
                                                ct := ct+1 ;
                                                printf("# n= %d m=%d r=%d itr=%d ct=%d\n",n,mextra,r,it
                                                alls := alls union {s} ;
                                                print(alls,nops(alls)) ;
                                                # printf("%d\n",s) ;
                                        else
                                                break ;
                                        end if;
                                        pell := expand(pell*qform) ;
                                end do:
                        end if;
                end if;
        end do:
        alls ;
end proc:

# generate all solutions for squares less than smax^2.
A175497(40000) ;
```

References

1. *Die lehre von den kettenbrüchen*, 3rd ed., Teubner Verlagsgesellschaft, Stuttgart, 1957.
2. Boaz Cohen, *Chebyshev polynomials and pell equations over finite fields*, Czech. Math. J. **71** (2021), no. 2, 491–510. MR 4263182
3. O. E. I. S. Foundation Inc., *The On-Line Encyclopedia Of Integer Sequences*, (2023), https://oeis.org/. MR 3822822
4. Jr H. W. Lenstra, *Solving the pell equation*, Not. Am. Math. Soc. **49** (2002), no. 2, 182–192. MR 1875156
5. Michael J. Jacabson and Hugh C. Williams, *Solving the pell equation*, Canad. Math. Soc., 2009. MR 2466979
6. Refik Kesin and Merve Güney Dunab, *Positive integer solutions of some pell equations*, Palest. J. Math. **8** (2019), no. 2, 213–226.
7. K. Matthews, *The diophantine equation* $x^2 - dy^2 = n$, Exposit. Math. **18** (2000), 323–331. MR 1788328
8. Keith Matthews, *Thue's theorem and the diophantine euqation* $x^2 - dy^2 = \pm n$, Mathem. Comput. **71** (2002), no. 239, 1281–1286. MR 1898757
9. J. McLaughlin, *Polynomial solutions of pell's equation and fundamental units in real quadratic fields*, J. London Math. Soc. **67** (2003), 16–28.
10. R. A. Mollin, *Simple continued fraction solutions for diophantine equations*, Expos. Mathem. **19** (2001), no. 1, 55–73. MR A1820127
11. Natl. Inst. Stand. Technol., *Digital library of mathematical functions*, NIST, 2022. MR 1990416
12. Ivan Niven, *Quadratic diophantine equations in the rational and quadratic fields*, Trans. Am. Math. Soc **52** (1942), 1–11. MR 0006739
13. Vladimir Pletser, *Triangular numbers multiple of triangular numbers and solutions of pell equations*, arXiv:2102.13494 [math.NT] (2021).
14. Bal Bahadur Tamang, *On the study of quadratic diophantine equations*, Master's thesis, Dept. Math. Inst. Sci. Techn. Tribhuvan Univ. Kathmandu Nepal, 2021.
15. Ahmet Tekcan, *Continued fractions equation of* $\sqrt{d}$ *and the pell equation* $x^2 - dy^2 = 1$, Mathem. Moravica **15** (2011), no. 2, 19–27.

URL: `http://www.mpia.de/~mathar`

Hoeschstr. 7, 52372 Kreuzau, Germany